
A Uniform Approach for Modeling
Electrical Machines

Michael Beuschel

Abstract

In this paper, an approach is presented that enables
uniform modeling of different types of electrical ma-
chines using a novel Modelica library of magnetic
components. Results of simulations with Dymola are
presented. This approach is also applicable to educa-
tion.

1 Introduction

Modelica provides a very general approach of model-
ing physical systems. Libraries for electrical and elec-
tronic as well as for mechanical components are al-
ready distributed on an open source code basis.

Based on this, a new library1 for modeling rotational
magnetic fields has been developed also including in-
terfaces to electrical and mechanical components.

In this paper, rotary electro-magnetic motors are
considered. Basically, different types of electrical ma-
chines employ the same physical principle: A mag-
netic field is produced that always tends towards a state
of minimal energy. This is achieved by a change of the
rotor position, which is the desired effect of a motor.

2 A Modelica Library of Magnetic
Components

Calculation of magnetic circuits is often done using a
notation similar to that of electrical circuits. There-
fore, themagnetic fluxψ and themagnetic potential
difference∆θ can be treated like currenti and voltagev
respectively. These are used in the following to model
magnetic components.

2.1 Magnetic Connectors

As the focus of this paper is on modeling rotating
electrical machines, both magnetic fluxψx, ψy and

1For details on the Modelica implementation, please see the
appendix.

magnetic potential difference∆θx, ∆θy are used in
2-dimensional space vector representation including
real and imaginary part (x- and y-axis respectively).
This is reflected by the definition ofmagnetic connec-
tors MagPandMagN, which only have different icons
to identify more easily the pins of a component (see
Fig. 1).

2.2 Basic Magnetic Components

Some basic magnetic components have been imple-
mented (see Fig. 1).

� As in electrical circuits, amagnetic ground
(MagGround ) is mandatory in every magnetic
circuit model to define the “magnetic potential”
for simulation.2

θx = 0

θy = 0 (1)

� A permanent magnet is amagnetic source(Mag-
Source ), that generates a magnetic potential
difference∆θ with angular orientationβ.

∆θx = ∆θ cos(β)
∆θy = ∆θ sin(β) (2)

� A linear magnetic resistance(MagResis-
tance ) connects the magnetic potential differ-
ence with the magnetic flux by

ψxRm = ∆θx

ψyRm = ∆θy (3)

with Rm = N2=M

The magnetic resistanceRm is determined by the
numberN of turns and the corresponding (electri-
cal) inductanceM. For 2-dimensional simulation,
the above operation is calculated for the real and
imaginary part of the magnetic field separately.



Figure 1: Basic magnetic components

Employing basic magnetic components, interfaces
to electrical and mechanical components are discussed
in the next section (see Fig. 2).

Figure 2: Magnetic interface components

2.3 Magnetic Coupling

A linear magnetic coupling(MagCoupling ) is based
on the electricalOnePort class. It relates electrical
voltagev and currenti to magnetic potential difference
and flux due to the induction law. An additional scal-
ing factork adjusts magnetic to electrical values due to
simplified modeling and field geometry.

v = �
N
k

�
cos(β)

dψx

dt
+ sin(β)

dψy

dt

�
(4)

∆θx = kN i cos(β)
∆θy = kN i sin(β) (5)

N is the number of turns;β gives the orientation of
the winding. Combining a magnetic coupling with a
magnetic resistance (3), the well known equationv=
�M di=dt of an inductance is obtained.

2.4 Commutator

A commutatorblock (Commutator ) is based on the
block magnetic coupling with additional rotation of
the magnetic field orientation due to the function of
a commutator in DC machines.

As the winding of a rotor moves forward the mag-
netic field rotates backwards in the rotor coordinate

2In physics no magnetic monopole is known. Thus, the “mag-
netic potential”θ ist only used in the magnetic ground and the
magnetic connector classes to distinguish from the magnetic po-
tential difference∆θ that always needs two reference points.

system. This is why the negative mechanical angle
�Zϕ is used here instead ofβ. Z scales the me-
chanical angle to obtain the magnetic one, whereZ
is half the number of poles. The mechanical connec-
tor flange b can be connected to components of the
Modelica.Mechanics.Rotational library.

2.5 Stator and Rotor

To model the interaction between the stationary and
rotational part of electrical machines, astator rotor
block (StatorRotor ) is employed. It provides
transformation between stator and rotor coordinates
and calculates the mechanical torqueτ from magnetic
flux ψx, ψy and potential difference∆θx, ∆θy.

0 = ψ1x + ψ2x cos(Zϕ) � ψ2y sin(Zϕ)
0 = ψ1y + ψ2x sin(Zϕ) + ψ2y cos(Zϕ) (6)

∆θ1x = ∆θ2x cos(Zϕ) � ∆θ2y sin(Zϕ)
∆θ1y = ∆θ2x sin(Zϕ) + ∆θ2y cos(Zϕ) (7)

τ = �Zψ2x∆θ2y + Zψ2y ∆θ2x (8)

3 Modeling Electrical Machines Us-
ing the Magnetics Library

The components of the magnetics library have been
tested implementing common electrical machines in
the Dymola simulation environment. In Figure 3 the
corresponding icons of a DC machine, a permanent
magnet DC machine, an induction AC machine and a
permanent magnet synchronous AC machine are dis-
played.

Figure 3: Electrical machine models (icons)

3.1 DC machine

Figure 4 shows the implementation of the DC ma-
chine. The stator winding provides the flux due to
the stator currentIS throughPositivePin1and Nega-
tivePin1. The related magnetic field is applied to the



Figure 4: DC machine implementation

stator rotor block at an angle of 90Æ using a magnetic
coupling block.

The magnetic resistanceRm is determined by the in-
ductanceLR and the numberNR of turns of the rotor
winding, which also has to match the corresponding
commutator block.

The induced voltage (back emf) is calculated em-
ploying (4). Whereas the orientation of the flux is con-
stant in stator coordinates, from the view point of the
rotor coordinate system it rotates. This is achieved by
introducing a commutator block. The number of poles
(2Z) has to be identical for the stator rotor block as
well as for the commutator, of course.

Employing the fluxψS= ∆θS=Rm= LSIS=NS of the
stator winding and the magnetic potential difference
∆θR= IRNR caused by the rotor currentIR throughPos-
itivePin2 and NegativePin2, the torqueτ the and in-
duced voltagevi of the DC machine can be calculated
employing the machine constantkm= 2ZNR=π.

τ =
2
π
�ZψSNRIR = kmψS�

∆θR

NR
(9)

vi =
2
π
�ZψSNRω = kmψSω (10)

Assuming an ideal DC machine, the fluxψS is almost
equally spread over 180Æ=Z of the airgap. In the same
way, also the rotor currentIR is the same for alle turns
(depending on the type of the rotor winding).

However, as the magnetics library employs a space
vector representation of the flux, a scaling factork=

Figure 5: DC machine simulation results

2=π for the flux has to be introduced in the magnetic
coupling block in order to model (9) and (10) correctly.
Hence, the magnetic potential difference∆θy and, ap-
plying the magnetic ressistanceRm, the magnetic flux
ψy are:

∆θy =
2
π

∆θS

ψy =
2
π

ψS

An acceleration procedure of this machine using an-
other current controller for the rotor currentIR as well
as a speed controller and applying a load torque of
20Nmhas been simulated, see Fig. 5.

The data of the implemented DC machine are as
follows:

Inductance Stator LS = 6.4 H
Inductance Rotor LR = 4.0 mH
Turns Stator Winding NS = 2400
Turns Rotor Winding NR = 60
Resistance Stator RS = 1.0 Ω
Resistance Rotor RR = 0.25 Ω
Number of Poles / 2 Z = 4
Mass Inertia J = 0.43 kgm2

3.2 Permanent Magnet DC Machine

The DC machine has then been modified using a per-
manent magnet to provide the flux at an angle of 90Æ

(see Fig. 6). The magnetic potential difference∆θy of
the magnetic source is set to get the same fluxψy as
above. The scaling factork is included, too.

∆θy =
2
π
� ISNS (11)



The same acceleration procedure as in Fig. 5 has been
simulated, see Fig. 7.

Figure 6: Permanent magnet DC machine implemen-
tation

Figure 7: Permanent magnet DC machine simulation
results

3.3 Simulation of an Induction AC Machine

Employing components of the magnetics library, also
an induction AC machine has been implemented (see
Fig. 8). The three stator windings are modeled sepa-
rately, including resistanceRS and leakage inductance
LSσ each. They are coupled to the magnetic circuit us-
ing magnetic coupling blocks at angular orientation of
0Æ, 120Æ and 240Æ.

The magnetic resistanceRm may either be applied
to the stator or to the rotor side of the magnetic cir-
cuit (the first one is chosen here). The magnetic resis-
tance is determined by the mutual inductanceM and
the numberNS of turns in each stator winding, which
has to match the number of turns in the corresponding
magnetic coupling blocks.

Figure 8: Induction AC machine implementation

Commonly, the mutual inductanceM is referred to a
single phase and is defined using the amplitude of the
flux vectorj~ψj and the peak phase currentbIS.

M = NS
j~ψjbIS (12)

Figure 9: Space vector representation of stator current
IS

The amplitude of the magnetic potential difference
vectorj∆~θj is (see also Fig. 9)���∆~θ��� = NS

���ISa(t)+ ISb(t)ej 120Æ + ISc(t)ej 240Æ
���

= N
3
2
bIS (13)

which determines the actual magnetic resistanceRm

usingj~ψj andj∆~θj as

Rm =

���∆~θ���
j~ψj

=
NS

3
2
bIS

M
NS

bIS =
3
2
�
N2

S

M
(14)



The same calculation is applied to the leakage induc-
tancesLSσ andLRσ.3

The rotor has to employ at least two windings at
equally spaced angle. In the example in Fig. 8, a 3-
phase rotor winding is modeled.

A start-up of this machine connected to symmetric
3-phase mains (ve f f = 230V , f = 50Hz) and applying
a load torque of 20Nm has been simulated, see
Fig. 10. The data of the implemented induction AC
machine are as follows (all numbers of turns equal 1):

Leakage Inductance StatorLSσ = 2.1 mH
Leakage Inductance RotorLRσ = 1.9 mH
Mutual Inductance M = 32.2 mH
Resistance Stator RS = 324 mΩ
Resistance Rotor RR = 203 mΩ
Number of Poles / 2 Z = 3
Mass Inertia J = 0.8 kgm2

Figure 10: Induction AC machine simulation results

3.4 Permanent Magnet Synchronous AC Ma-
chine

Based on the above induction machine, a synchronous
AC machine has been simulated, where the rotor flux
is provided by a permanent magnet (see Fig. 11). The
stator is identical to the induction machine. At the ro-
tor, a 2-pole damping winding has been introduced to
obtain a smooth torque output without vector control.

3For an improved version of the magnetics library, 3 magnetic
resistances related to the 3 windings should be employed rather
than a single inductance. However, this would require angle sen-
sitive magnetic resistances that are not yet implemented. Alter-
natively, the leakage inductances can also be implemented in the
electrical circuits.

Figure 11: Synchronous AC machine implementation

Figure 12 shows simulation results of the syn-
chronous AC machine. The stator windings have
been connected to a frequency and amplitude sweep 3-
phase supply. The magnetic source applies a magnetic
potential difference∆θ = NSIS= 5A that corresponds
to a flux of ψ = ∆θ=Rm = 1:71Vs. The data of the
damping windings of the implemented machine are as
follows:

Leakage Inductance RotorLRσ = 1.0 mH
Resistance Rotor RR = 40 mΩ

Figure 12: Synchronous AC machine simulation re-
sults

4 Conclusion

A new Modelica library of magnetic components has
been implemented and tested simulating DC and AC



machines. Simulation results have been validated us-
ing conventional motor models.

The presented approach enables a uniform and in-
tuitive modeling of different types of electrical ma-
chines. It also shows that different types of electrical
machines employ the same basic principles. There-
fore, this approach might be attractive especially for
education purposes.

However, due to redundant model variables com-
pared to conventional models, the presented approach
is not optimized in terms of simulation efficiency. It
also appears to be numerically more sensitive. There-
fore, the quality of simulation results significantly de-
pends on the integration algorithm and its tolerance
setting.

5 Outlook

Further investigations should be done regarding the
magnetics library itself as well as its application.

A variable magnetic resistance, a nonlinear mag-
netic resistanceRm(∆θ) and a magnetic resistance
Rm(β) with angular orientation should be introduced
to enable modeling of e.g. saturation, variable air gap
and reluctance effects (e.g. switched reluctance mo-
tors). Furthermore, the existing components can be
improved employing a vector implementation. This
would extend the library to 3-dimensional modeling
(e.g. of magnetic bearings).

The presented models of electrical machines can
then be refined and extended, e.g. by modeling leak-
age effects by individual magnetic components. In
addition, also other magnetic devices such as 1-phase
and 3-phase transformers can be modeled employing
the magnetics library.

Appendix:
Modelica Package ”Magnetics”

This package will become available on the Modelica
homepagehttp://www.Modelica.org/library/library.html.

package Magnetics

connector MagP "Positive magnetic pin"
SIunits.MagneticPotentialDifference theta_x;
SIunits.MagneticPotentialDifference theta_y;
flow SIunits.MagneticFlux psi_x;
flow SIunits.MagneticFlux psi_y;

end MagP;

connector MagN "Negative magnetic pin"
SIunits.MagneticPotentialDifference theta_x;
SIunits.MagneticPotentialDifference theta_y;
flow SIunits.MagneticFlux psi_x;
flow SIunits.MagneticFlux psi_y;

end MagN;

class MagGround "Magnetic ground"
Modelica.Electrical.Analog.Magnetics.MagP mag_p;

equation
mag_p.theta_x = 0;
mag_p.theta_y = 0;

end MagGround;

class MagSource "Magnetic potential difference source"
parameter SIunits.Angle beta=1e-8;
parameter SIunits.MagneticPotentialDifference theta=1;
SIunits.MagneticPotentialDifference theta_x;
SIunits.MagneticPotentialDifference theta_y;
Modelica.Electrical.Analog.Magnetics.MagP mag_p;
Modelica.Electrical.Analog.Magnetics.MagN mag_n;

equation
theta_x = mag_p.theta_x - mag_n.theta_x;
theta_y = mag_p.theta_y - mag_n.theta_y;
0 = mag_p.psi_x + mag_n.psi_x;
0 = mag_p.psi_y + mag_n.psi_y;
theta_x = theta*cos(beta);
theta_y = theta*sin(beta);

end MagSource;

class MagResistance "Magnetic resistance"
parameter Real N(final min=0) = 1;
parameter SIunits.Inductance M = 1;
SIunits.MagneticPotentialDifference theta_x;
SIunits.MagneticPotentialDifference theta_y;
SIunits.MagneticFlux psi_x;
SIunits.MagneticFlux psi_y;



Modelica.Electrical.Analog.Magnetics.MagP mag_p;
Modelica.Electrical.Analog.Magnetics.MagN mag_n;

equation
theta_x = mag_p.theta_x - mag_n.theta_x;
theta_y = mag_p.theta_y - mag_n.theta_y;
0 = mag_p.psi_x + mag_n.psi_x;
0 = mag_p.psi_y + mag_n.psi_y;
psi_x = mag_p.psi_x;
psi_y = mag_p.psi_y;
N*N*psi_x = M*theta_x;
N*N*psi_y = M*theta_y;

end MagResistance;

class MagCoupling "Linear magnetic coupling"
extends Modelica.Electrical.Analog.Interfaces.OnePort;
parameter SIunits.Angle beta = 1e-8 "Mag. Field Orient.";
parameter Real N(final min=0) = 1 "Number of Turns";
parameter Real k(final min=0) = 1 "Scaling Factor";
SIunits.MagneticPotentialDifference theta_x;
SIunits.MagneticPotentialDifference theta_y;
SIunits.MagneticFlux psi_x;
SIunits.MagneticFlux psi_y;
Modelica.Electrical.Analog.Magnetics.MagP mag_p;
Modelica.Electrical.Analog.Magnetics.MagN mag_n;

equation
theta_x = mag_p.theta_x - mag_n.theta_x;
theta_y = mag_p.theta_y - mag_n.theta_y;
0 = mag_p.psi_x + mag_n.psi_x;
0 = mag_p.psi_y + mag_n.psi_y;
psi_x = mag_p.psi_x;
psi_y = mag_p.psi_y;
v = -N/k*cos(beta)*der(psi_x) - N/k*sin(beta)*der(psi_y);
theta_x = N*k*i*cos(beta);
theta_y = N*k*i*sin(beta);

end MagCoupling;

class Commutator "Commutator with magnetic coupling"
extends Modelica.Electrical.Analog.Interfaces.OnePort;
parameter Real Z(final min=0) = 1 "Number of Poles / 2";
parameter Real N(final min=0) = 1 "Number of Turns";
parameter Real k(final min=0) = 1 "Scaling Factor";
SIunits.Angle phi "Rotational Magnetic Angle";
SIunits.MagneticPotentialDifference theta_x;
SIunits.MagneticPotentialDifference theta_y;
SIunits.MagneticFlux psi_x;
SIunits.MagneticFlux psi_y;
Modelica.Electrical.Analog.Magnetics.MagP mag_p;
Modelica.Electrical.Analog.Magnetics.MagN mag_n;
Modelica.Mechanics.Rotational.Interfaces.Flange_b flange_b;

equation
theta_x = mag_p.theta_x - mag_n.theta_x;
theta_y = mag_p.theta_y - mag_n.theta_y;
0 = mag_p.psi_x + mag_n.psi_x;
0 = mag_p.psi_y + mag_n.psi_y;
psi_x = mag_p.psi_x;
psi_y = mag_p.psi_y;
0 = flange_b.tau;
phi = -flange_b.phi*Z;
v = -N/k*cos(phi)*der(psi_x) - N/k*sin(phi)*der(psi_y);
theta_x = N*k*i*cos(phi);
theta_y = N*k*i*sin(phi);

end Commutator;

class StatorRotor "Stator and rotor of electric machines"
parameter Real Z(final min=0) = 1 "Number of Poles / 2";
SIunits.Angle phi(final start=1e-8) "Rotational Angle";
SIunits.MagneticPotentialDifference theta_1x "port 1";
SIunits.MagneticPotentialDifference theta_1y "port 1";
SIunits.MagneticPotentialDifference theta_2x "port 2";
SIunits.MagneticPotentialDifference theta_2y "port 2";
SIunits.MagneticFlux psi_1x "port 1";
SIunits.MagneticFlux psi_1y "port 1";
SIunits.MagneticFlux psi_2x "port 2";
SIunits.MagneticFlux psi_2y "port 2";

equation
theta_1x = mag_1p.theta_x - mag_1n.theta_x;
theta_1y = mag_1p.theta_y - mag_1n.theta_y;
theta_2x = mag_2p.theta_x - mag_2n.theta_x;
theta_2y = mag_2p.theta_y - mag_2n.theta_y;
0 = mag_1p.psi_x + mag_1n.psi_x;
0 = mag_1p.psi_y + mag_1n.psi_y;
0 = mag_2p.psi_x + mag_2n.psi_x;
0 = mag_2p.psi_y + mag_2n.psi_y;
psi_1x = mag_1p.psi_x;
psi_1y = mag_1p.psi_y;
psi_2x = mag_2p.psi_x;
psi_2y = mag_2p.psi_y;
flange_b.tau = Z*psi_2x*theta_2y - Z*psi_2y*theta_2x;
phi = flange_b.phi*Z;
0 = psi_1x + psi_2x*cos(phi) - psi_2y*sin(phi);
0 = psi_1y + psi_2x*sin(phi) + psi_2y*cos(phi);
theta_1x = theta_2x*cos(phi) - theta_2y*sin(phi);
theta_1y = theta_2x*sin(phi) + theta_2y*cos(phi);

end StatorRotor;

class DC_machine "DC machine using magnetic elements"
parameter Real Z(final min=0) = 4 "Number of Poles / 2";

parameter Real L_rotor(final min=0) = 0.004;
parameter Real N_stator(final min=0) = 2400;
parameter Real N_rotor(final min=0) = 60;
parameter Real R_stator(final min=0) = 1;
parameter Real R_rotor(final min=0) = 0.25;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin1;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin1;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin2;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin2;

Modelica.Electrical.Analog.Basic.Resistor
Resistor_Stator(R=R_stator);

Modelica.Electrical.Analog.Basic.Resistor
Resistor_Rotor(R=R_rotor);

Modelica.Electrical.Analog.Magnetics.MagCoupling
MagCoupling(beta=3.14159/2, N=N_stator, k=2/3.14159);

Modelica.Electrical.Analog.Magnetics.Commutator
Commutator(Z=Z, N=N_rotor);

Modelica.Electrical.Analog.Magnetics.MagGround
MagGround_Stator;

Modelica.Electrical.Analog.Magnetics.MagGround
MagGround_Rotor;

Modelica.Electrical.Analog.Magnetics.StatorRotor
StatorRotor(Z=Z);

Modelica.Electrical.Analog.Magnetics.MagResistance
MagResistance(N=N_rotor, M=L_rotor);

Modelica.Mechanics.Rotational.Interfaces.Flange_b
Flange_b;

equation
connect(PositivePin1, Resistor_Stator.p);
connect(PositivePin2, Resistor_Rotor.p);
connect(NegativePin2, Commutator.n);
connect(NegativePin1, MagCoupling.n);
connect(Resistor_Stator.n, MagCoupling.p);
connect(Resistor_Rotor.n, Commutator.p);
connect(MagCoupling.mag_p, StatorRotor.mag_1p);
connect(MagCoupling.mag_n, StatorRotor.mag_1n);
connect(MagGround_Stator.mag_p, MagCoupling.mag_n);
connect(MagGround_Rotor.mag_p, Commutator.mag_n);
connect(StatorRotor.mag_2p, MagResistance.mag_p);
connect(MagResistance.mag_n, Commutator.mag_n);
connect(Commutator.mag_p, StatorRotor.mag_2n);
connect(StatorRotor.flange_b, Flange_b);
connect(Commutator.flange_b, Flange_b);

end DC_machine;

class DC_PM_machine "Permanent magnet DC machine"
parameter Real Z(final min=0) = 4 "Number of Poles / 2";
parameter Real Theta_stator(final min=0) = 36000*2/3.14159

"Stator mag. Pot. Diff.";
parameter Real L_rotor(final min=0) = 0.004;
parameter Real R_rotor(final min=0) = 0.25;
parameter Real N_rotor(final min=0) = 60;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin1;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin1;

Modelica.Electrical.Analog.Basic.Resistor
Resistor_Rotor(R=R_rotor);

Modelica.Electrical.Analog.Magnetics.MagSource
MagSource(beta=3.14159/2, theta=Theta_stator);

Modelica.Electrical.Analog.Magnetics.StatorRotor
StatorRotor(Z=Z);

Modelica.Electrical.Analog.Magnetics.MagGround
MagGround_Stator;

Modelica.Electrical.Analog.Magnetics.MagGround
MagGround_Rotor;

Modelica.Electrical.Analog.Magnetics.MagResistance
MagResistance(N=N_rotor, M=L_rotor);

Modelica.Electrical.Analog.Magnetics.Commutator
Commutator(Z=Z, N=N_rotor,
psi_y(start=-Theta_stator*L_rotor/N_rotorˆ2));

Modelica.Mechanics.Rotational.Interfaces.Flange_b
Flange_b;

equation
connect(StatorRotor.mag_2p, MagResistance.mag_p);
connect(Resistor_Rotor.n, Commutator.p);
connect(PositivePin1, Resistor_Rotor.p);
connect(Commutator.n, NegativePin1);
connect(MagResistance.mag_n, Commutator.mag_n);
connect(MagGround_Stator.mag_p, StatorRotor.mag_1n);
connect(MagGround_Rotor.mag_p, Commutator.mag_n);
connect(Commutator.mag_p, StatorRotor.mag_2n);
connect(StatorRotor.flange_b, Flange_b);
connect(Commutator.flange_b, Flange_b);
connect(MagSource.mag_p, StatorRotor.mag_1p);
connect(MagSource.mag_n, MagGround_Stator.mag_p);

end DC_PM_machine;

class AC_machine ÄC induction machine"
parameter Real Z(final min=0) = 3 "Number of Poles / 2";
parameter Real M_mutual(final min=0) = 0.0322;
parameter Real L_stator_leakage(final min=0) = 0.0021;



parameter Real L_rotor_leakage(final min=0) = 0.0019;
parameter Real N_stator(final min=0) = 1 "Stator turns";
parameter Real N_rotor(final min=0) = 1 "Rotor turns";
parameter Real R_stator(final min=0) = 0.324;
parameter Real R_rotor(final min=0) = 0.203;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin1;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin1;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin2;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin2;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin3;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin3;

Modelica.Electrical.Analog.Basic.Resistor
Resistor1(R=R_stator);

Modelica.Electrical.Analog.Basic.Resistor
Resistor2(R=R_stator);

Modelica.Electrical.Analog.Basic.Resistor
Resistor3(R=R_stator);

Modelica.Electrical.Analog.Basic.Resistor
Resistor4(R=R_rotor);

Modelica.Electrical.Analog.Basic.Resistor
Resistor5(R=R_rotor);

Modelica.Electrical.Analog.Basic.Resistor
Resistor6(R=R_rotor);

Modelica.Electrical.Analog.Basic.Ground Ground4;
Modelica.Electrical.Analog.Basic.Ground Ground5;
Modelica.Electrical.Analog.Basic.Ground Ground6;
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling1(beta=0, N=N_stator);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling2(beta=2*3.14159265/3, N=N_stator);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling3(beta=4*3.14159265/3, N=N_stator);
Modelica.Electrical.Analog.Magnetics.MagGround

MagGround_Stator;
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling4(beta=0, N=N_rotor);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling5(beta=2*3.14159265/3, N=N_rotor);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling6(beta=4*3.14159265/3, N=N_rotor);
Modelica.Electrical.Analog.Magnetics.MagGround

MagGround_Rotor;
Modelica.Electrical.Analog.Magnetics.MagResistance

MagResistance(N=N_stator, M=M_mutual*2/3);
Modelica.Electrical.Analog.Magnetics.MagResistance

MagResistance1(N=N_stator, M=L_stator_leakage*2/3);
Modelica.Electrical.Analog.Magnetics.MagResistance

MagResistance2(N=N_rotor, M=L_rotor_leakage*2/3);
Modelica.Electrical.Analog.Magnetics.StatorRotor

StatorRotor(Z=Z);
Modelica.Mechanics.Rotational.Interfaces.Flange_b

Flange_b1;
equation

connect(PositivePin1, Resistor1.p);
connect(NegativePin1, MagCoupling1.n);
connect(PositivePin2, Resistor2.p);
connect(NegativePin2, MagCoupling2.n);
connect(PositivePin3, Resistor3.p);
connect(NegativePin3, MagCoupling3.n);
connect(MagCoupling2.mag_n, MagCoupling3.mag_p);
connect(MagCoupling3.mag_n, MagGround_Stator.mag_p);
connect(MagCoupling4.n, Resistor4.n);
connect(MagCoupling6.n, Resistor6.n);
connect(MagCoupling5.n, Resistor5.n);
connect(MagCoupling4.mag_n, MagCoupling5.mag_p);
connect(MagCoupling5.mag_n, MagCoupling6.mag_p);
connect(MagCoupling6.n, Ground6.p);
connect(MagCoupling5.n, Ground5.p);
connect(MagCoupling4.n, Ground4.p);
connect(MagGround_Rotor.mag_p, MagCoupling6.mag_n);
connect(MagCoupling1.mag_p, MagResistance.mag_p);
connect(MagGround_Stator.mag_p, StatorRotor.mag_1n);
connect(StatorRotor.mag_2p, MagCoupling4.mag_p);
connect(MagGround_Rotor.mag_p, StatorRotor.mag_2n);
connect(MagCoupling1.mag_n, MagCoupling2.mag_p);
connect(MagResistance.mag_n, StatorRotor.mag_1p);
connect(MagCoupling4.p, Resistor4.p);
connect(MagCoupling5.p, Resistor5.p);
connect(MagCoupling6.p, Resistor6.p);
connect(StatorRotor.flange_b, Flange_b1);
connect(MagResistance1.mag_p, MagResistance.mag_p);
connect(MagResistance1.mag_n, MagGround_Stator.mag_p);
connect(Resistor3.n, MagCoupling3.p);
connect(Resistor2.n, MagCoupling2.p);
connect(Resistor1.n, MagCoupling1.p);
connect(MagResistance2.mag_p, StatorRotor.mag_2p);
connect(MagResistance2.mag_n, MagGround_Rotor.mag_p);

end AC_machine;

class AC_PM_machine ÄC PM machine using magnetic elements"
parameter Real Z(final min=0) = 3 "Number of Poles / 2";
parameter Real Theta_rotor(final min=0) = 0.172*2/3.14159;

parameter Real M_mutual(final min=0) = 0.0322;
parameter Real L_stator_leakage(final min=0) = 0.0021;
parameter Real L_rotor_leakage(final min=0) = 0.001;
parameter Real N_stator(final min=0) = 1 "Stator turns";
parameter Real N_rotor(final min=0) = 1 "Rotor turns";
parameter Real R_stator(final min=0) = 0.324;
parameter Real R_rotor(final min=0) = 0.04;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin1;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin1;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin2;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin2;

Modelica.Electrical.Analog.Interfaces.NegativePin
NegativePin3;

Modelica.Electrical.Analog.Interfaces.PositivePin
PositivePin3;

Modelica.Electrical.Analog.Basic.Resistor
Resistor1(R=R_stator);

Modelica.Electrical.Analog.Basic.Resistor
Resistor2(R=R_stator);

Modelica.Electrical.Analog.Basic.Resistor
Resistor3(R=R_stator);

Modelica.Electrical.Analog.Basic.Resistor
Resistor4(R=R_rotor);

Modelica.Electrical.Analog.Basic.Resistor
Resistor5(R=R_rotor);

Modelica.Electrical.Analog.Basic.Ground Ground4;
Modelica.Electrical.Analog.Basic.Ground Ground6;
Modelica.Electrical.Analog.Magnetics.MagSource

MagSource(beta=0, theta=Theta_rotor);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling1(beta=0, N=N_stator);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling2(beta=2*3.14159265/3, N=N_stator);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling3(beta=4*3.14159265/3, N=N_stator);
Modelica.Electrical.Analog.Magnetics.MagCoupling

MagCoupling4(beta=0, N=N_rotor,
psi_x(start=Theta_rotor*M_mutual/N_rotorˆ2*2/3));

Modelica.Electrical.Analog.Magnetics.MagCoupling
MagCoupling5(beta=3.14159265/2, N=N_rotor);

Modelica.Electrical.Analog.Magnetics.MagGround
MagGround_Stator;

Modelica.Electrical.Analog.Magnetics.MagGround
MagGround_Rotor;

Modelica.Electrical.Analog.Magnetics.MagResistance
MagResistance(N=N_stator, M=M_mutual*2/3);

Modelica.Electrical.Analog.Magnetics.MagResistance
MagResistance1(N=N_stator, M=L_stator_leakage*2/3);

Modelica.Electrical.Analog.Magnetics.MagResistance
MagResistance2(N=N_rotor, M=L_rotor_leakage);

Modelica.Electrical.Analog.Magnetics.StatorRotor
StatorRotor(Z=Z);

Modelica.Mechanics.Rotational.Interfaces.Flange_b
Flange_b1;

equation
connect(PositivePin1, Resistor1.p);
connect(NegativePin1, MagCoupling1.n);
connect(PositivePin2, Resistor2.p);
connect(NegativePin2, MagCoupling2.n);
connect(PositivePin3, Resistor3.p);
connect(NegativePin3, MagCoupling3.n);
connect(MagCoupling2.mag_n, MagCoupling3.mag_p);
connect(MagCoupling3.mag_n, MagGround_Stator.mag_p);
connect(MagCoupling4.n, Resistor4.n);
connect(MagCoupling5.n, Resistor5.n);
connect(MagCoupling5.n, Ground6.p);
connect(MagCoupling4.n, Ground4.p);
connect(MagGround_Rotor.mag_p, MagCoupling5.mag_n);
connect(MagCoupling1.mag_p, MagResistance.mag_p);
connect(MagGround_Stator.mag_p, StatorRotor.mag_1n);
connect(StatorRotor.mag_2p, MagCoupling4.mag_p);
connect(MagGround_Rotor.mag_p, StatorRotor.mag_2n);
connect(MagCoupling1.mag_n, MagCoupling2.mag_p);
connect(MagResistance.mag_n, StatorRotor.mag_1p);
connect(MagCoupling4.p, Resistor4.p);
connect(MagCoupling5.p, Resistor5.p);
connect(StatorRotor.flange_b, Flange_b1);
connect(MagResistance1.mag_p, MagResistance.mag_p);
connect(MagResistance1.mag_n, MagGround_Stator.mag_p);
connect(Resistor3.n, MagCoupling3.p);
connect(Resistor2.n, MagCoupling2.p);
connect(Resistor1.n, MagCoupling1.p);
connect(MagResistance2.mag_p, StatorRotor.mag_2p);
connect(MagResistance2.mag_n, MagGround_Rotor.mag_p);
connect(MagCoupling4.mag_n, MagSource.mag_p);
connect(MagSource.mag_n, MagCoupling5.mag_p);

end AC_PM_machine;

end Magnetics;


